

Operadores Zonais em Álgebra de Mapas e Sua Aplicação a Zoneamento Ecológico-Econômico

CLÁUDIO CLEMENTE BARBOSA¹
GILBERTO CAMARA¹
JOSÉ SIMEÃO DE MEDEIROS¹
EDSON CREPANI¹
EVLYN NOVO¹
JOÃO PEDRO CERVEIRA CORDEIRO¹

¹INPE--Instituto Nacional de Pesquisas Espaciais
Caixa Postal 515, 12201- 097 São José dos Campos, SP, Brasil
{claudio, gilberto, simeao, jpedro}@dpi.inpe.br
{evlyn, crepani} @ltid.inpe.br

Abstract. This paper describes the concept of zonal operators for Map Algebra, its implementation in the SPRING system, and the application of this class of operators to the partial automatization of procedures for ecological-economical zoning.

Keywords: Geoprocessing, Map Algebra, Zonal Operators, Ecological-Economical Zoning.

1 Introdução

Uma parcela significativa dos projetos de aplicações ambientais de Geoprocessamento utiliza técnicas de integração de dados básicos de várias fontes e formatos. Como exemplo, os procedimentos metodológicos propostos pelo INPE e LAGET/UFRJ para ZEE-Zoneamento Ecológico-Econômico (Crepani et al., 1996) envolvem o uso de imagens de satélite, mapas pedológicos, geológicos, de geomorfologia, de vegetação e de clima, com a geração de produtos intermediários, como cartas de vulnerabilidade à erosão e a integração dos dados do meio físico-biótico a dados sócio-econômicos.

Dadas a complexidade e a dimensão dos procedimentos envolvidos em projetos como o ZEE, torna-se muito importante desenvolver ferramentas de Geoprocessamento que possam auxiliar estes empreendimentos, em especial ao automatizar tarefas trabalhosas e repetitivas. Para este fim, muitos sistemas de informação geográfica (SIG) incluem uma linguagem de manipulação de dados espaciais, que permite definir procedimentos complexos de integração de dados. Exemplos de tais linguagens são GRID (ARC/INFO) e LEGAL (SPRING).

Em trabalhos anteriores (Câmara, 1995; Cordeiro et al., 1996) discutimos os principais conceitos da linguagem LEGAL e apresentamos um conjunto de *operadores pontuais*, com exemplos de aplicação. Este trabalho discute os conceitos e a implementação em LEGAL de um conjunto adicional de operadores, denotados na literatura como *zonais*, que permitem estabelecer relações espaciais entre *regiões* de mapas distintos, distintamente dos operadores pontuais, que relacionam *pontos* individuais de diferentes mapas. Mostramos ainda um exemplo de aplicação

desta classe de operadores para a semi-automatização de procedimentos para zoneamento ecológico-econômico.

2 Operadores Zonais em Álgebra de Mapas

O termo “álgebra de mapas” foi cunhado por Tomlin (1990), para indicar o conjunto de procedimentos de análise espacial em Geoprocessamento que produz novos dados, a partir de funções de manipulação aplicadas a um ou mais mapas. Esta visão concebe a análise espacial como um conjunto de operações matemáticas sobre mapas, em analogia aos ambientes de álgebra e estatística tradicional. Os mapas são tratados como variáveis individuais, e as funções definidas sobre estas variáveis são aplicadas de forma homogênea a todos os pontos do mapa.

A Álgebra de Mapas compõe uma linguagem especializada para realizar operações que tem tanto um sentido matemático quanto cartográfico e espacial. Estas operações podem ser agrupadas em três grandes classes:

1. *Pontuais*: a saída da operação é um mapas cujos valores são função apenas dos valores dos mapa de entrada em cada localização correspondente. Podem operar apenas sobre um mapa(e.g. *fatiar* um modelo numérico de terreno) ou realizar intersecções entre conjuntos espaciais (e.g. *operações booleanas* entre mapas temáticos)¹.
2. *Vizinhança*: o resultado é um mapa cujos valores dependem da vizinhança da localização considerada. Exemplos são a *filtragem* espacial de uma imagem e o cálculo de *declividade* de um MNT.
3. *Zonais*: definidas sobre regiões específicas de um mapa de entrada, onde as restrições são fornecidas por outro mapa. Um exemplo seria: “dado um mapa de solos e um mapa de declividade da mesma região, obtenha a declividade média para cada tipo de solo”.

As operações de transformações zonais ou por região operam sobre um mapa de origem e um conjunto de regiões delimitadas por polígonos na forma matricial ou vetorial. Diferentemente das transformações de vizinhança, onde cada posição geográfica possui sua própria vizinhança representada por uma máscara que se desloca sobre os dados, nas transformações zonais as regiões são estáticas, ou seja, não se deslocam sobre a região geográfica de estudo.

As operações zonais mais comuns, segundo Tomlin (1990), são:

- **MÁXIMO ZONAL** e **MÍNIMO ZONAL**: Gera um mapa onde o valor resultante em todas as posições geográficas que compõem uma região é igual ao maior valor ou ao menor valor do atributo encontrado na mesma região sobre o mapa de origem.
- **MAIORIA ZONAL**: No resultado, o valor em todas as posição geográficas dentro de uma região é igual ao valor de maior frequência encontrado na mesma região sobre o mapa de origem.

¹Estas operações podem envolver modificação da topologia (e.g. uma reclassificação é usualmente combinada com uma junção topológica).

- **MÉDIA ZONAL:** O valor produzido, em todas as localizações que compõem uma região, é igual à média aritmética dos valores da variável geográfica nas mesmas localizações no mapa de origem.
- **DIVERSIDADE ZONAL:** Gera um resultado onde o valor resultante em todas as posições geográficas que compõem uma região é igual ao número de valores diferentes do atributo no mapa de origem.
- **ESTATÍSTICA ZONAL:** Produz uma estatística dos valores para cada região. O resultado é um dado tabular, indicando, por exemplo para cada região, os valores máximo, médio, mínimo, variância, desvio padrão e diversidade.

Como exemplo, considere-se a operação MAIORIA ZONAL, onde o mapa de origem é um mapa de uso da terra, e as regiões de restrição, são definidas por um mapa temático de tipos de solo. O tema de uso da terra de maior área dentro de uma região preencherá todo o espaço da região no mapa resultante.




Figura 1 - Operação de Maioria zonal (Barbosa, 1997).

3 A Importância do Conceito de Zonalidade em Geoprocessamento

Como apresentado na seção anterior, o conceito de zonalidade é utilizado em operadores de análise espacial, cujas restrições espaciais são definidas por áreas. A importância teórica deste conceito é dupla: permite materializar num SIG os conceitos de unidade de paisagem (Bertrand, 1981; Tricart, 1977; Tricart e KiewietdeJonge, 1992) e área-unidade (Hartshorne, 1978) e permite a ligação entre dados do meio físico-biótico e dados sócio-econômicos, essencial para estudos de ordenação do território.

Na visão de Hartshorne, uma área-unidade (*unit-area*) é uma partição do espaço geográfico, definida pelo pesquisador em função do objeto de estudo e da escala de trabalho, que apresenta características individuais próprias; estas áreas-unidades seriam a base de um sistema de classificação e organização do espaço. A partir da decomposição do espaço em áreas-unidade, o

pesquisador poderá relacionar, para cada uma destas partições, as correspondentes características físicas-bióticas que a individualizam em relação a todas as demais componentes do espaço.

Por exemplo, na metodologia para Zoneamento Ecológico-Econômico descrita em Crepani et al. (1996) e Becker e Egler (1996), o uso de imagens de satélite serve como base para definição de unidades de paisagem (chamadas *unidades territoriais básicas*). Uma *unidade territorial básica* (UTB) exprime o conceito geográfico de zonalidade através de atributos ambientais que permitem diferenciá-la de outras unidades vizinhas, ao mesmo tempo em que possui vínculos dinâmicos que a articulam a uma complexa rede integrada por outras unidades territoriais. Estas UTBs são definidas por foto-interpretação em imagens de satélite e servem como âncora para a definição dos mapas derivados do meio físico-biótico.

Outro uso importante dos operadores zonais em Geoprocessamento é no relacionamento entre dados do meio físico e dados sócio-econômico. Enquanto os dados do meio físico são usualmente expressos sob forma de mapas temáticos e modelos numéricos de terreno, os dados sócio-econômicos estão normalmente agrupados em setores censitários, e organizados segundo um mapa cadastral, com atributos num banco de dados relacional. Deste modo, a espacialização destes setores censitários corresponde a polígonos no mapa. Tais polígonos delimitam regiões sobre as quais se pode computar operações zonais sobre mapas básicos ou derivados do meio físico.

Como ilustração do uso operadores zonais para integração entre dados físico-bióticos e sócio-econômico, tomemos a operação “dados o mapa de municípios e os mapas de precipitação média mensal para o Estado de São Paulo, calcule a precipitação média mensal para cada município”.

4 Operações Zonais em LEGAL

O sistema SPRING (Câmara et al., 1996) inclui uma linguagem de manipulação e consulta de mapas (LEGAL), que vem sendo implementada a partir de proposta original definida em Câmara (1995). As principais inovações, com relação a linguagens tradicionais de álgebra de mapas, *no contexto de operações zonais*, incluem:

- Linguagens de Álgebra de Mapas como GRID (ARC/INFO) e PCRaster (McDonnell and Burrough, 1998) utilizam apenas dados em formato matricial (*raster*), representação computacional que é natural para operações matemáticas entre diferentes mapas, mas que impõe a necessidade de conversão entre formatos. Em LEGAL, não há necessidade de converter o mapa de referência (que define as regiões) da representação matricial para a vetorial, o que simplifica o procedimento e melhora sua acurácia.
- As operações zonais em LEGAL possuem uma interface direta com tabelas de um bancos de dados relacional. Deste modo, o resultado de uma operação zonal pode ser diretamente armazenado como um atributo de elementos do banco de dados.

Como exemplo, considere o estudo, desenvolvido por Matteo (1998), para determinar o relacionamento entre o valor do índice de vegetação normalizado (“NDVI”), obtido das bandas TM4 e TM3 de uma imagem LANDSAT, com a produtividade de talhões de uma fazenda de

cana de açúcar. A imagem estava no formato matricial e os talhões estavam associados a um mapa cadastral, com atributos armazenados num banco de dados relacional. O programa em LEGAL que obtem, para cada talhão, a média dos valores de NDVI, é apresentado na Figura 2. Os comandos iniciados por duas barras verticais indicam comentários do programa.

```
// Parte 1 - Definicao de dados
Objeto      talhoes ("Talhoes1996");
Cadastral   safra96 ("MapadeTalhoes");
Imagem      ndvi ("LANDSAT");
// Parte 2 - Recuperacao dos Dados
safra96 = Recupere (Nome = "safra96");
ndvi     = Recupere (Nome = "ndvi");
// Parte 3 - Operacao Zonal
talhoes."NDVI" = MediaZonal (ndvi, talhoes OnMap safra96);
```

Figura 2 – Programa em LEGAL que integra dados matriciais e atributos de bancos de dados (Fonte: Matteo (1998).)

A última linha do programa realiza, de uma só vez, todo o procedimento desejado. O campo “NDVI”, correspondente a um atributo de cada talhão, será preenchido com os valores médios do índice de vegetação, calculados sobrepondo-se na imagem de satélite, os limites da cada talhão.

Não conhecemos, na literatura especializada, exemplo de linguagem para Álgebra de Mapas que consiga executar procedimentos que combinem mapas, bancos de dados e operadores zonais com semelhante grau de concisão.

4 Aplicação em Zoneamento Ecológico-Econômico

4.1 Adaptação da Metodologia do ZEE

Para exemplificar a utilização dos operadores zonais em LEGAL, escolhemos uma aplicação complexa: a geração de uma carta temática de vulnerabilidade natural à erosão, parte do procedimento de caracterização do meio físico nos processos de Zoneamento Ecológico Econômico(ZEE), conforme metodologia desenvolvida pelo INPE e pelo LAGET/UFRJ (Crepani et al.,1996) e (Becker e Egler,1996). O roteiro metodológico para elaboração destas cartas segue os seguintes passos, tal como adaptado para uso com LEGAL:

1. Elaboração de um mapa preliminar de unidades homogêneas de paisagem obtidas a partir da análise e interpretação visual de LANDSAT/TM, considerando os padrões fotográficos identificados pela variação dos matizes de cores, e pelos elementos texturais de relevo e drenagem.

2. Associação de cada um dos mapas base de Geologia, Geomorfologia, Pedologia e Cobertura Vegetal a pesos que indicam a contribuição relativa de cada tema, para os processos de morfogênese e pedogênese. A partir de cada mapa temático, serão gerados modelos numéricos de terreno nos quais os valores estarão entre o mínimo de 1 (estabilidade com predomínio da pedogênese) e 3 (instabilidade, com predomínio da morfogênese). Por exemplo, no caso da Geologia, áreas com rochas resistentes à erosão são caracterizadas com valores próximos de 1.
3. Geração de mapas derivados, através de uma operação zonal entre o mapa de unidades territoriais básicas (UTB), obtido na etapa (1) com o modelo numérico de terreno resultante da ponderação de cada mapa temático, obtido na etapa (2). Esta etapa deverá produzir novos modelos numéricos, com a distribuição das contribuições de cada componente do meio físico esteja homogeneizada pela zonalidade das UTBs.
4. Realizar uma operação de média ponderada entre os mapas gerados na etapa (3), o que permitirá integrar a contribuição de cada componente do meio físico para as diferentes UTBs. O dado resultante será um único modelo numérico, com valores entre 1 e 3.
5. Proceder a um fatiamento do modelo resultante, gerando assim uma carta temática de vulnerabilidade natural a erosão.

4.2 Validação da Metodologia

Para validar a adaptação da metodologia, escolhemos uma área de trabalho que já havia sido interpretada por métodos tradicionais, o que permitiu uma base de comparação com o resultado obtido. A região de estudo corresponde à folha Rio Fresco, código SB-22-Y-D e está situada no estado do Pará, perfazendo uma área de 18150 Km² (165km x 110km), e esta localizada entre os paralelos 7° 00' e 8° 00' de latitude sul e os meridianos 51° 00' e 52° 30' de longitude oeste. O mapa de localização da região é apresentado na figura 6.1. Os dados utilizados incluem:

- Imagens do satélite TM/LANDSAT -5, bandas 3,4,5, na escala 1:250000. Estas imagens foram utilizadas tanto no formato digital quanto em uma composição colorida (5R4G3B) em papel. Para cobrir toda a área foi necessário fazer um mosaico de duas cenas do satélite (órbita 224, ponto 65, de 24/07/1992 e órbita 225, ponto 65. Data 31/07/1992).
- Quatro cartas temáticas na escala 1:1000000 do Projeto RADAMBRASIL. Volume 4- Folhas SB/SC 22, com o Mapa Geológico, Mapa FitoEcológico, Mapa Geomorfológico e Mapa Exploratório de Solos.
- Carta topográfica na escala 1:250000, confeccionada pelo Instituto Brasileiro de Geografia e Estatística (IBGE/1982). Projeção UTM/SAD-69, carta Rio Fresco, Folha SB-22-Y-D MIR224.

A metodologia descrita na seção 4.1 foi utilizada, com o programa resultante em LEGAL apresentado em anexo. A carta de vulnerabilidade natural à erosão obtida está mostrada na Figura 3. A comparação entre os resultados obtidos de forma manual com os produzidos automaticamente mostrou um alto grau de concordância. As diferenças que ocorreram foram ocasionadas por problemas nos mapas básicos originais. Uma análise mais detalhada dos mapas temáticos do RADAM indicou que há divergências relevantes entre os diferentes mapas,

especialmente entre os mapas geológicos e os geomorfológicos. Assim, em projetos de integração de dados, é imprescindível realizar um pre-processamento nos mapas temáticos básicos, para corrigir eventuais imprecisões relativas existentes.

5 Comentários Finais

Neste trabalho, discutimos a concepção e implementação dos operadores de Álgebra de Mapas sobre regiões (*operadores zonais*) no contexto da linguagem LEGAL. Demonstramos ainda a viabilidade do uso destes operadores para a realização de procedimentos complexos de análise espacial, como os utilizados no Zoneamento Ecológico-Econômico. Os resultados mostraram que, desde que os mapas básicos sejam corrigidos para garantir a correlação espacial entre os diferentes temas, os operadores zonais de LEGAL são ferramentas poderosas e de fácil utilização em aplicações ambientais que exigem integração de dados.




Figura 3. Mapa de Vulnerabilidade Natural da Região de Rio Fresco (PA).

Referências

Barbosa, C.C.F *Álgebra de Mapas e Suas Aplicações em Sensoriamento Remoto e Geoprocessamento*. Tese de Mestrado em Sensoriamento Remoto, INPE, 1996 (disponível em <http://www.dpi.inpe.br/teses/claudio>).

- Becker, B. K.; Egler, C. A. G. *Detalhamento da Metodologia para Execução do Zoneamento Ecológico-Econômico pelos Estados da Amazônia Legal*. Brasília. SAE-Secretaria de Assuntos Estratégicos/ MMA-Ministério do Meio Ambiente. 1996.
- Bertrand, G. *Paisagem e Geografia Física Global: Esboço Metodológico*. Caderno de Ciências da Terra, 13. Inst. de Geografia-USP. São Paulo. 1981.
- Câmara, G. *Modelos, Linguagens e Arquiteturas para Bancos de Dados Geográficos*. Tese de Doutorado em Computação Aplicada. São José dos Campos, INPE, 1995 (disponível na Internet em <http://www.dpi.inpe.br/people/gilberto>)
- Câmara, G.; Souza, R.C.M.; Freitas, U.M.; Garrido, J.C.P. "SPRING: Integrating Remote Sensing and GIS with Object-Oriented Data Modelling". *Computers and Graphics*, vol.15 , n.6, July 1996, pp.13-22.
- Cordeiro, J.P.S.; Amaral, S.; Freitas, U.M.; Câmara, G. Álgebra de Geo-Campos e Suas Aplicações. In: Simpósio Brasileiro de Sensoriamento Remoto, 1996. *Anais*, São José dos Campos, INPE, 1996 (<http://www.dpi.inpe.br/people/gilberto>).
- Crepani, E.; Medeiros, J. S.; Azevedo, L. G.; Hernandez, P.; Florenzano, T. G.; Duarte, V. *Curso de Sensoriamento Remoto Aplicado ao Zoneamento Ecológico-Econômico*. São José dos Campos. INPE. 1996.
- Hartshorne, R. *Propósitos e Natureza da Geografia*. São Paulo, HUCITEC/EDUSP, 1978.
- Matteo, K. C. *Sistemas de Informação Geográfica para Monitoramento da Cultura de Cana de Açúcar*. Tese de Mestrado em Sensoriamento Remoto, INPE, 1998.
- McDonell, R.; Burrough, P.A. *Principles of Geographical Information Systems*. Oxford, Oxford University Press, 1998.
- Tomlin, D. *Geographic information systems and Cartographic Modeling*. Prentice Hall, New York, 1990.
- Tricart, J. *Ecodinâmica*. Rio de Janeiro. IBGE-SUPREN,1977.
- Tricart, J.; KiewietdeJonge, C. *Ecogeography and Rural Management: A Contribution to the International Geosphere-Biosphere Programme*. Essex. Longman Scientific & Technical. 1992.

ANEXO

PROGRAMA EM LEGAL PARA GERAÇÃO DO MAPA DE VULNERABILIDADE NATURAL À EROSÃO

```
{
// Inicio do programa
// Variáveis usadas para receber os planos de informação com
// mapas temáticos de entrada, usados durante a manipulação.
Tematico MapaSolo ("pedologia");
```



```

Tematico MapaGeologia ("geologia");
Tematico MapaVegetacao ("vegetacao");
Tematico MapaGeomorfologia ("geomorfologia");
Tematico MapaUTB ("utbasica");

// Variáveis numéricas usadas para integração de dados Temáticos
Digital SoloPonderado ("Numerico");
Digital GeologiaPonderado ("Numerico");
Digital VegetacaoPonderado ("Numerico");
Digital GeomorfologiaPonderado ("Numerico");

// Variáveis tabela usadas para integração de dados Temáticos
Tabela TbNotasSolos (Ponderacao);
Tabela TbNotasGeologia (Ponderacao);
Tabela TbNotasVegetacao (Ponderacao);
Tabela TbNotasGeomorfologia (Ponderacao);

// Variáveis Numéricas usadas para receber os mapas
// com o valor médio do tema em cada unidade territorial básica.
// (representação grade regular de pontos)
Digital MediaUTBSolo ("Numerico");
Digital MediaUTBGeologia ("Numerico");
Digital MediaUTBVegetacao ("Numerico");
Digital MediaUTBGeomorfologia ("Numerico");
Digital VulnerabilidadeNumerica ("Numerico");

// Variável usada para Mapa temático de vulnerabilidade
Tematico VulnerabilidadeTematica ("vulnerabilidade");

// Tabela usada na operação pontual de fatiamento da grade
// regular de pontos com valores médios de vulnerabilidade.
Tabela TbVulnerabilidade (Fatiamento);

// Ações de instânciação dos dados
// Associação dos dados de entrada com as variáveis declaradas
MapaSolo = Recuperere (Nome= "solo_atualizado");
MapaGeologia = Recuperere (Nome= "go_atualizado");
MapaVegetacao = Recuperere (Nome= "veg_atualizado");
Mapageomorfologia = Recuperere (Nome= "gm_atualizado");
MapaUTB = Recuperere (Nome= "utbas");

// Criação dos mapas usados para integrar os dados tematicos
// segundo um modelo que estabelece 21

```

```

// classes de vulnerabilidade (valores de 1.0 a 3.0)
SoloPonderado      =      Novo(Nome="solo_ponderado_60",      ResX=60,      ResY=60,
Escala=100000, Min=0, Max=3);
GeologiaPonderado  =      Novo(Nome="geo_ponderado_60",      ResX=60,      ResY=60,
Escala=100000, Min=0, Max=3);
VegetacaoPonderado =      Novo(Nome="veget_ponderado_60",    ResX=60,      ResY=60,
Escala=100000, Min=0, Max=3);
GeomorfologiaPonderado = Novo(Nome="geomorf_ponderado_60", ResX=60, ResY=60,
Escala=100000, Min=0, Max=3);

// Criação dos mapas usados para
// receber os valores médios de cada tema para cada utb
MediaUTBSolo      =      Novo(Nome="mediautbasicasolo_60",    ResX      = 60, ResY=60,
Escala=100000, Min=0, Max=3);
MediaUTBGeologia  =      Novo(Nome="mediautbasicageologia_60", ResX=60, ResY=60,
Escala=100000, Min=0, Max=3);
MediaUTBVegetacao =      Novo(Nome="mediautbasicavegetacao_60", ResX=60, ResY=60,
Escala=100000, Min=0, Max=3);
MediaUTBGeomorfologia = Novo( Nome      = "mediautbageomorfol_60", ResX=60,
ResY=60, Escala=100000, Min=0, Max=3);

// Criação dos mapas usados para
// receber o valor médio final de cada utb
VulnerabilidadeNumerica = Novo(Nome="vulnerabilidadeNum", ResX=60, ResY=60,
Escala=100000, Min=0, Max=3);

// Criação do mapa tematico final de vulnerabilidade

VulnerabilidadeTematica = Novo(Nome="vulnerabilidade", ResX=60, ResY=60,
Escala=100000);

// Criação da tabela a ser usada para integração das informações
// provenientes do tema solo.

TbNotasSolos = Novo ( CategoriaIni="pedologia",
"PB1" : 2.0, "PB5" : 2.0, "HG" : 3.0, "R1" : 3.0, "TR" : 2.0, "R2" : 3.0,
"LV7" : 1.0, "LV8" : 1.0, "ar" : 3.0, "A" : 3.0);

// Criação da tabela a ser usada para integração das informações
// provenientes do tema geologia
TbNotasGeologia = Novo ( CategoriaIni="geologia",
"pEx_c_xingu" : 1.5, "pEgp_g_grao_para" : 1.4, "pErn_f_rio_fresco" : 2.7,
"pEac_gran_s_carajas" : 1.3,
"pEse_f_sobrei" : 1.9, "pEia_form_Iriri" : 1.0, "pEav_gran_v_guilher" : 1.2,
"pEgo_f_gorotire" : 2.6, "Qa_Qc_aluviao" : 3.0 );

```

```

// Criação da tabela a ser usada para integração das informações
// provenientes do tema vegetação
TbNotasVegetacao = Novo ( CategoriaIni="vegetacao",
"Fdt" : 1.0, "arFdt" : 3.0, "Fam" : 1.3, "Fal" : 1.2,
"Sc" : 1.7, "Scmd" : 1.8, "Srmd" : 2.2, "Sr" : 2.0 );

// Criação da tabela a ser usada para integração das informações
// provenientes do tema geomorfologia
TbNotasGeomorfologia = Novo ( CategoriaIni="geomorfologia",
"SEpt" : 1.2, "Estb" : 1.3, "Esp" : 1.1, "Ei" : 1.9,
"Egi" : 2.2, "dr" : 2.3, "drv" : 2.7, "dk" : 2.4, "dkr" : 2.6,
"dke" : 2.5, "drt" : 2.1, "dkit" : 2.0, "dm" : 1.2,
"dmit" : 1.4, "dcta" : 1.6, "dci" : 2.0, "atf" : 1.0,
"apf" : 3.0, "dit" : 1.4, "dc" : 1.6 );

// Criação da tabela a ser usada no fatiamento da grade
// regular de pontos com as notas médias de vulnerabilidade.
TbVulnerabilidade = Novo (CategoriaFim = "vulnerabilidade",
[1.0,1.0999] : "estavel", [1.0999,1.1999] : "estavel1",
[1.1999,1.2999] : "estavel2", [1.2999,1.3999] : "estavel3",
[1.3999,1.4999] : "moderadamente_estavel",
[1.4999,1.5999] : "moderadamente_estavel1",
[1.5999,1.6999] : "moderadamente_estavel2",
[1.6999,1.7999] : "moderadamente_estavel3",
[1.7999,1.8999] : "medianamente_estavel",
[1.8999,1.9999] : "medianamente_est_vul",
[1.9999,2.0999] : "medianamente_est_e_ou_vul",
[2.0999,2.1999] : "medianamente_vul_est",
[2.1999,2.2999] : "medianamente_vulneravel",
[2.2999,2.3999] : "moderadamente_vul1",
[2.3999,2.4999] : "moderadamente_vul2",
[2.4999,2.5999] : "moderadamente_vul3",
[2.5999,2.6999] : "moderadamente_vul",
[2.6999,2.7999] : "vulneravel1",
[2.7999,2.8999] : "vulneravel2",
[2.8999,2.9999] : "vulneravel3",
[2.9999,3.0] : "vulneravel" );

// Operação de ponderação pontual das classes dos temas solo,
// geologia, geomorfologia e vegetação.

GeologiaPonderado = Pondere (MapaGeologia, Tbnotasgeologia);

```

```

GeomorfologiaPonderado = Pondere (MapaGeomorfologia,
                                TbNotasGeomorfologia);
SoloPonderado      = Pondere (MapaSolo, TbNotasSolos);
VegetacaoPonderado = Pondere (Mapavegetacao, TbNotasVegetacao);

// Determinação do valor médio em cada unid. territorial básica
// para o temas geologia, geomorfologia, solos e vegetacao
// (Operação de média zonal).
MediaUTBGeologia = MediaZonal (GeologiaPonderado,
                               MapaUTB.Classe * );
MediaUTBGeomorfologia = MediaZonal (GeomorfologiaPonderado,
                                    MapaUTB.Classe * );
MediaUTBSolo = MediaZonal (SoloPonderado,
                           MapaUTB.Classe * );
MediaUTBVegetacao = MediaZonal (VegetacaoPonderado,
                                MapaUTB.Classe * );

// Determinação da nota média final de vulnerabilidade de cada
// unidade territorial básica

VulnerabilidadeNumerica = ( MediaUTBGeologia + MediaUTBGeomorfologia +
                           MediaUTBSolo + MediaUTBVegetacao ) /4;

// Operação de fatiamento da grade regular com as notas
// médias de vulnerabilidade.
VulnerabilidadeTematica = Fatie ( VulnerabilidadeNumerica,
                                TbVulnerabilidade);
} // Fim do programa

```